Pensieve header: A unified verification notebook for the PPSA project.

Continues pensieve://2017-06/ and pensieve://2017-08/.

Prolog

```
Go;
```

```
SetOptions[EvaluationNotebook[],
  NotebookEventActions → {
    {"MenuCommand", "Save"} :→ (
      SetDirectory["C:\\drorbn\\AcademicPensieve\\Projects\\PPSA"];
      Get["MakeVSnips.m"]
     ),
    PassEventsDown → True
   }];
```

```
HL[ℰ_] := Style[ℰ, Background → Yellow];
```

Initialization

The "degree carrier / filtration parameter" is ℏ, and all "coupling constants" are proportional to it.

TD

```
$TℏD = 5; ℏ /: ℏ^d_. /; d > $TℏD := 0;
$TϵD = 3; ϵ /: ϵ^d_. /; d > $TϵD := 0;
SetAttributes[SS, HoldAll];
SS[ℰ_] := Block[{ℏ, ϵ},
  Normal@Series[ℰ, {ℏ, 0, $TℏD}, {ϵ, 0, $TϵD}] ]
Simp[ℰ_] := Collect[ℰ, _CU | _QU, Expand];
```

DeclareAlgebra

QLImplementation

```
Unprotect[NonCommutativeMultiply];
NonCommutativeMultiply[x_] := x; 0 ** _ = _ ** 0 = 0;
B[x_, x_] = 0; B[x_, y_] := x ** y - y ** x;
(x_Plus) ** y_ := (# ** y) & /@ x; x_ ** (y_Plus) := (x ** #) & /@ y;
```

QLImplementation

```
DeclareAlgebra[U_Symbol, opts__Rule] := Module[{gp, sr, cp, S, pow,
    gs = Generators /. {opts}, cs = Centrals /. {opts}
  },
  gp = Alternatives @@ gs;
  gp = gp | gp_; (* generators pattern *)
  sr = Thread[gs → Range@Length@gs]; (* sorting rule *)
  cp = Alternatives @@ cs; (* centrals pattern *)
  S[ℰ_] := Collect[ℰ, _U, Expand];
  U_i_[ℰ_] := ℰ /. {t : cp ⧴ t_i, u_U ⧴ Replace[u, x_ ⧴ x_i, 1]};
  B[U@(x_)_i_, U@(y_)_i_] := B[U@x_i, U@y_i] = U_i@B[U@x, U@y];
  B[U@(x_)_i_, U@(y_)_j_] /; i =!= j := 0;
  B[U@y_, U@x_] := S[-B[U@x, U@y]];
  x_ ** U[] := x; U[] ** x_ := x;
  (a_ * x_U) ** (b_ * y_U) := If[a b === 0, 0, S[a b (x ** y)]];
  (a_ * x_U) ** y_ := S[a (x ** y)]; x_ ** (a_ * y_U) := S[a (x ** y)];
  U[xx___, x_] ** U[y_, yy___] := If[OrderedQ[{x, y} /. sr],
    U[xx, x, y, yy], U@xx ** (U@y ** U@x + B[U@x, U@y]) ** U@yy];
  U@{c_. * (l : gp)^n_, r___} /; FreeQ[c, gp] := S[c U@Table[l, {n}] ** U@{r}];
  U@{c_. * l : gp, r___} := S[c U[l] ** U@{r}];
  U@{c_, r___} /; FreeQ[c, gp] := S[c U@{r}];
  U@{} = U[];
  U@{l_Plus, r___} := S[U@{#, r} & /@ l];
  U@{l_, r___} := U@{Expand[l], r};
  𝕆_U[poly_, specs___] := Module[{sp, null, vs, us},
    sp = Replace[{specs}, l_List ⧴ (l → null), {1}];
    vs = Join @@ (First /@ sp);
    us = Join @@ (sp /. (l_ → s_) ⧴ (l /. x_i_ ⧴ x_s));
    S[Total[
        CoefficientRules[poly, vs] /. (p_ → c_) ⧴ c U@(us^p)
      ]] /. x_null ⧴ x
   ];
  pow[ℰ_, 0] = U[]; pow[ℰ_, n_] := pow[ℰ, n - 1] ** ℰ;
  𝕊_U[ℰ_, ss___Rule] := S@Total[
     CoefficientRules[ℰ, First /@ {ss}] /.
      (p_ → c_) ⧴ c NonCommutativeMultiply @@ MapThread[pow, {Last /@ {ss}, p}]]
  ]
```

DeclareMorphism

QLImplementation

```
DeclareMorphism[m_, U_ → V_, ongs_List, oncs_List: {}] := (
  Replace[ongs, (g_ → img_) ⧴ (m[U[g]] = img), {1}];
  m[U[]] = V[];
  m[U[g_i_]] := V_i[m[U@g]];
  m[U[vs__]] := NonCommutativeMultiply @@ (m /@ U /@ {vs});
  m[ℰ_] := Simp[ℰ /. oncs /. u_U ⧴ m[u]]; )
```

Implementing $sl_2^{\gamma\epsilon}$

CU

```
DeclareAlgebra[CU, Generators → {y, a, x}, Centrals → {t}];
B[CU@a, CU@y] = -γ CU@y; B[CU@x, CU@a] = -γ CU@x;
B[CU@x, CU@y] = 2 ϵ CU@a - t CU[];
```

Verifying associativity on triples of generators:

```
With[{bas = CU /@ {y, a, x}},
 Table[z1 ** (z2 ** z3) - (z1 ** z2) ** z3 // Simp,
  {z1, bas}, {z2, bas}, {z3, bas} ] ]
```

```
{{{0, 0, 0}, {0, 0, 0}, {0, 0, 0}},
 {{0, 0, 0}, {0, 0, 0}, {0, 0, 0}}, {{0, 0, 0}, {0, 0, 0}, {0, 0, 0}}}
```

Verifying associativity on a “random” triple:

```
With[{z1 = CU[y, y, a, a, x, x], z2 = CU[y, a, x], z3 = CU[y, y, a, x]}, {
   (rhs = (z1 ** z2) ** z3 // Simp) // Short,
   z1 ** (z2 ** z3) - rhs // Simp
  }] // Timing
```

$\{0.625,$
$\{(28\,t^2\gamma^4 + 116\,t\,\gamma^5\,\epsilon + 120\,\gamma^6\,\epsilon^2)\,\mathrm{CU}[y, y, y, x, x] + \ll 23 \gg + \mathrm{CU}[y, y, y, y, \ll 6 \gg, x, x, x], 0\}\}$

Implementing $\mathcal{U}_{\gamma\epsilon;\hbar}$

With $q = e^{\hbar\gamma\epsilon}$, $A = e^{-\hbar\epsilon a}$, and $[f, g]_q := fg - qgf$, our algebra is $\mathcal{U}_{\gamma\epsilon;\hbar} = \langle t, y, a, x\rangle/\mathcal{R}$, where
$\mathcal{R} = \left([t, *] = 0,\ [a, y] = -\gamma y,\ [x, y]_q = (1 - T^2 A^2)/\hbar,\ [x, a] = -\gamma x\right)$.

QU

```
DeclareAlgebra[QU, Generators → {y, a, x}, Centrals → {t, T}];
q := Normal@Series[ℯ^(γ ϵ ℏ), {ℏ, 0, $TℏD}];
B[QU@a, QU@y] = -γ QU@y; B[QU@x, QU@a] = -γ QU@x;
B[QU@x, QU@y] := (q - 1) QU@{y, x} + 𝕆_QU[SS[(1 - T^2 ℯ^(-2 ℏ ϵ a)) / ℏ], {a}];
```

```
With[{bas = QU /@ {y, a, x}}, Table[z1 ** z2 - z2 ** z1 // Simp, {z1, bas}, {z2, bas}] ]
```

$\left\{\left\{0, \gamma\,\mathrm{QU}[y], \left(-\frac{1}{\hbar} + \frac{T^2}{\hbar}\right)\mathrm{QU}[\,] - 2\,T^2\epsilon\,\mathrm{QU}[a] + 2\,T^2\epsilon^2\hbar\,\mathrm{QU}[a, a] + \left(-\gamma\,\epsilon\,\hbar - \frac{1}{2}\gamma^2\epsilon^2\hbar^2\right)\mathrm{QU}[y, x]\right\},\right.$
$\{-\gamma\,\mathrm{QU}[y], 0, \gamma\,\mathrm{QU}[x]\},$
$\left.\left\{\left(\frac{1}{\hbar} - \frac{T^2}{\hbar}\right)\mathrm{QU}[\,] + 2\,T^2\epsilon\,\mathrm{QU}[a] - 2\,T^2\epsilon^2\hbar\,\mathrm{QU}[a, a] + \left(\gamma\,\epsilon\,\hbar + \frac{1}{2}\gamma^2\epsilon^2\hbar^2\right)\mathrm{QU}[y, x], -\gamma\,\mathrm{QU}[x], 0\right\}\right\}$

Verifying associativity on triples of generators:

```
With[{bas = QU /@ {y, a, x}},
 Table[z1 ** (z2 ** z3) - (z1 ** z2) ** z3 // Simp,
  {z1, bas}, {z2, bas}, {z3, bas} ] ]
```

```
{{{0, 0, 0}, {0, 0, 0}, {0, 0, 0}},
 {{0, 0, 0}, {0, 0, 0}, {0, 0, 0}}, {{0, 0, 0}, {0, 0, 0}, {0, 0, 0}}}
```

Verifying associativity on a “random” triple (~34 secs):

```
With[{z1 = QU[y, y, a, a, x, x], z2 = QU[y, a, x], z3 = QU[y, y, a, x]}, {
   (rhs = (z1 ** z2) ** z3 // Simp) // Short,
   z1 ** (z2 ** z3) - rhs // Simp
  }] // Timing
```

$\{49.5938,$

$\left\{\left(135\,\gamma^6\,\epsilon^2 - 730\,T^2\,\gamma^6\,\epsilon^2 + \ll 8 \gg + \frac{198\,T^4\,\gamma^5\,\epsilon}{\hbar}\right) QU[y, y, y, x, x] + \ll 22 \gg + (\ll 1 \gg)\ \ll 1 \gg,\ 0\right\}\}$

Verifying that $\text{Lim}_{\hbar\to 0}$ QU = CU using a “random” product (~23 secs):

```
With[{z1 = CU[y, y, a, a, x, x], z2 = CU[y, a, x], z3 = CU[y, y, a, x]}, {
   Short[lhs = z1 ** (z2 ** z3)],
   Short[rhs = (QU @@ z1) ** ((QU @@ z2) ** (QU @@ z3))],
   Expand[Limit[rhs /. {QU → CU, T → ⅇ^(ℏ t/2)}, ℏ → 0] == lhs]
  }] // Timing
```

$\{28.0156,$

$\{2\,(8\,t^2\,\gamma^4 + 16\,t\,\gamma^5\,\epsilon)\,CU[y, y, y, x, x] + \ll 94 \gg + CU[y, y, y, y, y, a, a, a, a, x, x, x, x],$

$(-8\,T^2\,\gamma^6\,\epsilon^2 + 8\,T^4\,\gamma^6\,\epsilon^2)\,QU[y, y, y, x, x] + \ll 422 \gg + \left(\gamma\,\epsilon\,\hbar + \frac{15}{2}\,\gamma^2 \ll 1 \gg \hbar^2\right) \ll 1 \gg,\ \text{True}\}\}$

Implementing θ

theta

```
DeclareMorphism[Cθ, CU → CU, {y → -CU@x, a → -CU@a, x → -CU@y}, {t → -t, T → T^-1}];
DeclareMorphism[Qθ, QU → QU, {y → 𝕆_QU[SS[-T^-1 ⅇ^(ℏ ϵ a) x], {a, x}],
   a → -QU@a, x → 𝕆_QU[SS[-T^-1 ⅇ^(ℏ ϵ a) y], {a, y}]}, {t → -t, T → T^-1}]
```

Verifying involutivity on CU:

```
With[{bas = CU /@ {y, a, x}},
 Table[z → Cθ[z] → Cθ[Cθ[z]], {z, bas}] ]
```

```
{CU[y] → -CU[x] → CU[y], CU[a] → -CU[a] → CU[a], CU[x] → -CU[y] → CU[x]}
```

Verifying that θ is a multiplicative homomorphism on CU:

```
With[{bas = CU /@ {y, a, x}},
 Table[Cθ[z1 ** z2] - Cθ[z1] ** Cθ[z2], {z1, bas}, {z2, bas}] ]
```

```
{{0, 0, 0}, {0, 0, 0}, {0, 0, 0}}
```

Verifying involutivity on QU:

```
With[{bas = QU /@ {y, a, x}},
 Table[z → Qθ[z] → HL[Qθ[Qθ[z]]], {z, bas}]]
```

$$\left\{QU[y] \to -\frac{QU[x]}{T} - \frac{\epsilon\hbar\, QU[a,x]}{T} - \frac{\epsilon^2\hbar^2\, QU[a,a,x]}{2T} \to QU[y],\ QU[a] \to -QU[a] \to QU[a],\right.$$
$$\left. QU[x] \to \left(-\frac{1}{T} + \frac{\gamma\epsilon\hbar}{T} - \frac{\gamma^2\epsilon^2\hbar^2}{2T}\right) QU[y] + \left(-\frac{\epsilon\hbar}{T} + \frac{\gamma\epsilon^2\hbar^2}{T}\right) QU[y,a] - \frac{\epsilon^2\hbar^2\, QU[y,a,a]}{2T} \to QU[x]\right\}$$

Verifying that θ is a multiplicative homomorphism on QU:

```
With[{bas = QU /@ {y, a, x}},
 Table[{z1, z2} → Simp[Qθ[z1 ** z2] - Qθ[z1] ** Qθ[z2]], {z1, bas}, {z2, bas}]]
```

```
{{{QU[y], QU[y]} → 0, {QU[y], QU[a]} → 0, {QU[y], QU[x]} → 0},
 {{QU[a], QU[y]} → 0, {QU[a], QU[a]} → 0, {QU[a], QU[x]} → 0},
 {{QU[x], QU[y]} → 0, {QU[x], QU[a]} → 0, {QU[x], QU[x]} → 0}}
```

The Asymmetric Dequantizator

Following pensieve://People/VanDerVeen/Dequant1.pdf.

ADeq

$$\mathrm{A\mathbb{D}\$f} = \frac{\gamma}{\hbar} e^{\hbar\left(\frac{t}{2}-(a+\gamma)\epsilon\right)} \left(\left(\mathrm{Cosh}\left[\hbar\left(a\epsilon + \frac{\gamma\epsilon}{2} - \frac{t}{2}\right)\right] - \mathrm{Cosh}\left[\hbar\sqrt{\left(\frac{t-\gamma\epsilon}{2}\right)^2 + \epsilon\omega}\right]\right) \Big/ \left(\mathrm{Sinh}\left[\frac{\gamma\epsilon\hbar}{2}\right]\left(a^2\epsilon + a\gamma\epsilon - a t - \omega\right)\right)\right);$$

Scaling behaviour of A𝔻$f:

$$\mathrm{Simplify}\left[\mathrm{A\mathbb{D}\$f} == \left(\left(\mathrm{A\mathbb{D}\$f}\ /.\ \gamma\to 1\right)\ /.\ \left\{\epsilon \to \gamma\epsilon,\ a \to \gamma^{-1}a,\ \omega \to \gamma^{-1}\omega\right\}\right)\right]$$

True

$$\mathrm{FullSimplify}\left[\mathrm{A\mathbb{D}\$f} == \left(\left(\mathrm{A\mathbb{D}\$f}\ /.\ \gamma\to 1\right)\ /.\ \left\{\hbar \to \gamma^2\hbar,\ \epsilon \to \epsilon/\gamma,\ a \to a/\gamma,\ t \to \gamma^{-2}t,\ \omega \to \gamma^{-3}\omega\right\}\right)\right]$$

True

ADeq

$$\mathrm{A\mathbb{D}\$\omega} = \gamma\, CU[y,x] + \epsilon\, CU[a,a] - (t - \gamma\epsilon)\, CU[a];$$

ADeq

```
DeclareMorphism[A𝔻, QU → CU,
 {a → CU@a, x → CU@x, y → 𝕊CU[SS[A𝔻$f], a → CU[a], ω → A𝔻$ω] ** CU@y}]
```

Verifying that the asymmetric dequantizator is a homomorphism:

```
With[{bas = QU /@ {y, a, x}},
 Table[{z1, z2} → Simp[
    A𝔻[z1 ** z2] - A𝔻[z1] ** A𝔻[z2] /. {T → SS[ⅇ^(ℏ t/2)]}
   ], {z1, bas}, {z2, bas}] ]
```

$$\left\{\{\{QU[y], QU[y]\} \to 0, \{QU[y], QU[a]\} \to 0, \{QU[y], QU[x]\} \to 0\}, \{\{QU[a], QU[y]\} \to 0, \{QU[a], QU[a]\} \to 0, \{QU[a], QU[x]\} \to 0\}, \left\{\{QU[x], QU[y]\} \to \frac{t^6 \hbar^5 CU[]}{23040}, \{QU[x], QU[a]\} \to 0, \{QU[x], QU[x]\} \to 0\right\}\right\}$$

The Symmetric Dequantizator

Following pensieve://People/VanDerVeen/Dequant1.pdf.

SDeq

$$S\mathbb{D}\$P = \frac{\mathrm{Cosh}\left[\hbar\left(\frac{\epsilon-t}{2}+\epsilon\, a\right)\right]-\mathrm{Cosh}\left[\hbar\sqrt{\frac{t^2+\epsilon^2}{4}+\epsilon\,\varpi}\right]}{\hbar\,\mathrm{Sinh}\left[\frac{-\epsilon\hbar}{2}\right]\left(\varpi-\epsilon\, a^2+(t-\epsilon)\, a+t/2\right)}$$

SDeq

$$-\left(\left(\left(\mathrm{Cosh}\left[\left(a\,\epsilon+\frac{1}{2}(-t+\epsilon)\right)\hbar\right]-\mathrm{Cosh}\left[\sqrt{\frac{1}{4}\left(t^2+\epsilon^2\right)+\epsilon\,\varpi}\;\hbar\right]\right)\mathrm{Csch}\left[\frac{\epsilon\,\hbar}{2}\right]\right)\Big/\left(\left(\frac{t}{2}+a\,(t-\epsilon)-a^2\,\epsilon+\varpi\right)\hbar\right)\right)$$

```
Simplify[S𝔻$P == (S𝔻$P /. {a → -a - 1, t → -t})]
```

True

```
SS[S𝔻$P]
```

$$1+\left(\frac{t^2}{24}+\left(\frac{a}{12}+\frac{a^2}{12}\right)\epsilon^2+\epsilon\left(-\frac{t}{24}-\frac{a\,t}{12}+\frac{\varpi}{12}\right)\right)\hbar^2$$

SDeq

```
S𝔻$Q = Simplify[S𝔻$P /. {a → c - 1/2}]
```

SDeq

$$-\left(\left(4\left(\mathrm{Cosh}\left[\frac{1}{2}(t-2\,c\,\epsilon)\,\hbar\right]-\mathrm{Cosh}\left[\frac{1}{2}\sqrt{t^2+\epsilon^2+4\,\epsilon\,\varpi}\;\hbar\right]\right)\mathrm{Csch}\left[\frac{\epsilon\,\hbar}{2}\right]\right)\Big/\left(\left(4\,c\,t+\epsilon-4\,c^2\,\epsilon+4\,\varpi\right)\hbar\right)\right)$$

```
Simplify[S𝔻$Q == (S𝔻$Q /. {c → -c, t → -t})]
```

True

```
SS[SD$Q]
```

$$1+\left(\frac{t^2}{24}+\left(-\frac{1}{48}+\frac{c^2}{12}\right)\epsilon^2+\epsilon\left(-\frac{c\,t}{12}+\frac{\varpi}{12}\right)\right)\hbar^2$$

SDeq

```
SD$g = FullSimplify[√SD$Q /. c → a + 1/2]
```

SDeq

$$\sqrt{2}\sqrt{\left(\left(\left(-\mathrm{Cosh}\left[\frac{1}{2}\left(t-(1+2a)\epsilon\right)\hbar\right]+\mathrm{Cosh}\left[\frac{1}{2}\sqrt{t^2+\epsilon^2+4\epsilon\varpi}\,\hbar\right]\right)\mathrm{Csch}\left[\frac{\epsilon\hbar}{2}\right]\right)\Big/\left(\left(t+2at-2a(1+a)\epsilon+2\varpi\right)\hbar\right)\right)}$$

```
SS[SD$g]
```

$$1+\left(\frac{t^2}{48}+\left(\frac{a}{24}+\frac{a^2}{24}\right)\epsilon^2+\epsilon\left(-\frac{t}{48}-\frac{a\,t}{24}+\frac{\varpi}{24}\right)\right)\hbar^2$$

```
Simplify[SD$P == SD$g (SD$g /. {a → -a - 1, t → -t})]
```

True

SDeq

```
SD$f = FullSimplify[e^(-ϵ ℏ a) T (SD$g /. {a → -a, t → -t})]
```

SDeq

$$\sqrt{2}\,e^{-a\epsilon\hbar}\,T\sqrt{\left(\left(\left(-\mathrm{Cosh}\left[\frac{1}{2}\left(t+\epsilon-2a\epsilon\right)\hbar\right]+\mathrm{Cosh}\left[\frac{1}{2}\sqrt{t^2+\epsilon^2+4\epsilon\varpi}\,\hbar\right]\right)\mathrm{Csch}\left[\frac{\epsilon\hbar}{2}\right]\right)\Big/\left(\left((-1+2a)t-2(-1+a)a\epsilon+2\varpi\right)\hbar\right)\right)}$$

```
SS[SD$f]
```

$$T-aT\epsilon\hbar+\left(\frac{t^2T}{48}+\left(-\frac{aT}{24}+\frac{13a^2T}{24}\right)\epsilon^2+\epsilon\left(\frac{tT}{48}-\frac{atT}{24}+\frac{T\varpi}{24}\right)\right)\hbar^2$$

SDeq

```
SD$ϖ = γ CU[y, x] + ϵ CU[a, a] - (t - γ ϵ) CU[a] - t γ CU[] / 2;
```

SDeq

```
DeclareMorphism[SD, QU → CU, {a → CU@a,
  x → 𝕊CU[SS[SD$f /. {ℏ → γ^2 ℏ, ϵ → ϵ/γ, a → a/γ, t → γ^-2 t, ϖ → γ^-3 ϖ}],
     a → CU[a], ϖ → SD$ϖ] ** CU@x,
  y → 𝕊CU[SS[SD$g /. {ℏ → γ^2 ℏ, ϵ → ϵ/γ, a → a/γ, t → γ^-2 t, ϖ → γ^-3 ϖ}],
     a → CU[a], ϖ → SD$ϖ] ** CU@y
 }]
```

Verifying the θ-symmetry:

```
Table[Simplify[Cθ[S𝔻[z]] == S𝔻[Qθ[z]] /. {T^p_-· → SS[ℯ^(ℏ p t/2)]}], {z, QU /@ {y, a, x}}]
```

```
{True, True, True}
```

Verifying that the symmetric dequantizator is a homomorphism:

```
With[{bas = QU /@ {y, a, x}},
 Table[{z1, z2} → Simp[
    S𝔻[z1 ** z2] - S𝔻[z1] ** S𝔻[z2] /. {T^p_-· → SS[ℯ^(ℏ p t/2)]}
   ], {z1, bas}, {z2, bas}] ]
```

```
{{{QU[y], QU[y]} → 0, {QU[y], QU[a]} → 0, {QU[y], QU[x]} → 0},
 {{QU[a], QU[y]} → 0, {QU[a], QU[a]} → 0, {QU[a], QU[x]} → 0},
 {{QU[x], QU[y]} → 1/720 t^6 ℏ^5 CU[], {QU[x], QU[a]} → 0, {QU[x], QU[x]} → 0}}
```